

Using SAS® and Other Tools to Move an Institutional Research (IR) Office from Hardcopy Reporting to a Web-Based Environment

Sabrina Andrews, University of Central Florida, Orlando, FL
Evangeline Collado, University of Central Florida, Orlando, FL
Patricia Ramsey, University of Central Florida, Orlando, FL

ABSTRACT

At a minimum, university constituents want timely and accurate data. The focus now is the demand for user-friendly reporting. More specifically, in today's fast-paced, technology-enhanced world, users want a friendly, web-reporting environment. As these demands increase, it presents challenges for the IR office. At the University of Central Florida (UCF), a major metropolitan university, we are meeting some of these challenges by converting legacy-based reports into dynamic and interactive Web applications using SAS® tools, specifically SAS/IntrNet®, and other desktop applications.

We will demonstrate two applications:

- The *Enrollment Profile* uses SAS/IntrNet, Base SAS, SAS/GRAPH, SAS/Macro Language, HTML, JavaScript, and Microsoft Access to provide the university community with current and historical student enrollment headcount information.
- The *Course Registration Profile* uses the above tools and SAS/ACCESS to Oracle to provide the university administrative personnel, faculty, and staff with near-live course registration data during registration periods. Historical data are maintained for trend comparisons with previous year's data.

WHO WE ARE

MISSION

The mission of the Office of Institutional Research (IR) is to support planning, decision-making, and assessment at the University of Central Florida by providing information that is timely and of the highest quality.

PURPOSE

The IR office is a service-oriented unit that provides information and proposes alternative solutions to support the decision-making process. The purposes of the Office of Institutional Research are to assist in the organization and maintenance of institutional data for decision-making, to analyze and interpret data, to provide information which is appropriate and useful in planning and decision-making, and to serve as a catalyst for institutional self analysis and improvement. The director and staff serve on numerous university-wide committees and workgroups and assist with the collection and interpretation of institutional data, assist in planning academic programs, and participate in the implementation of evaluative procedures. The functions of the office support the entire university enterprise.

BACKGROUND

The Office of Institutional Research (IR) has, as one of its primary responsibilities, the task of reporting all official data to internal and external constituents. Staff from the IR office routinely meet with end users and participate in workgroups and committees that deal with data needs and information access. This allows us the opportunity to interact directly with end users and determine, one-on-one, exactly what their data needs are and the format that would best meet their needs. This also has the added benefit of allowing IR staff to more immediately respond to needs and make changes to our new dynamic applications as they are requested.

INTRODUCTION

Throughout the year, numerous official data files and hard copy

reports are produced. What we have found very recently is a trend that end-users want more and more data in varying formats. The legacy-based programs are cumbersome to change and the hard copy reports may not be as functional as the user needs. Having a more sophisticated and flexible reporting tool such as SAS/IntrNet has allowed the IR office to better respond to user's data and information needs.

UCF has been using SAS in the classroom for many years. Thus, IR was able to utilize the SAS knowledge of a member of the Statistics & Actuarial Science Department—Dr. Morgan Wang, Director of the SAS Data Mining Certificate program at UCF. With the assistance of several of his graduate students we began to explore applications built on SAS technology because of the ability to dynamically create and display tables and graphs or charts on the web. As a result of senior level administrators wanting more timely and easily accessible data it was decided that the IR office would lead an effort to use SAS in an "administrative capacity." Two applications were prototyped and will be discussed in this paper.

ENROLLMENT PROFILE

The Enrollment Profile site was originally conceived as a way for the IR office to be able to immediately respond to (daily) enrollment questions during key times of the year. The IR office was always the source for enrollment data and, before we had the capability to produce the data dynamically from a daily update process, we would have to wait until a hard copy printed out (often a day or two after the request was made). Since our census data file is not submitted until 6 weeks into the term, you can see where earlier data needed to be provided.

HARDCOPY REPORTING

Several hardcopy enrollment reports for each college and/or school at UCF are produced for each of an academic year's semesters—summer, fall and spring. A page of one hard to read, multiple-page report is shown in Figure 1. These reports were written in COBOL and are difficult to program and redesign. Based on space limitations, adding columns or rows would pose some challenges. One department may have 30, 40, or 50 pages in one reporting term and there are more than 60 departments at UCF. Thus, physical storage room to house these catalogued reports is at a premium and, if a college wanted to collect historical information for each of its departments, it would take a tremendous amount of time and effort just to photocopy that many pages. This prompted us to look for a better way to provide this information.

Figure 1: OCR-B2 Student Enrollment Survey Report

UNIVERSITY OF CENTRAL FLORIDA																			
OCR-B2 STUDENT ENROLLMENT SURVEY																			
PAGE: 27																			
DATE: 08/20/01																			
MAJOR FIELD OF STUDY: 99 9999 (999999) - SUMMARY REPORT																			
TERM: 08/2001																			
ALL STUDENTS ENROLLED FROM CHOICE	LINE	NON RESIDENT		BLACK		AMERICAN INDIAN OR ALASKAN NATIVE		ASIAN OR PACIFIC ISLANDER		HISPANIC		WHITE		NOT REPORTED		TOTAL		SUM OF COLUMNS	
		NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN	NEW	WOMEN
I. FULL-TIME STUDY TOTAL																			
A. UNDERGRAD TOTAL																			
1. FIRST YEAR																			
2. SECOND YEAR																			
3. THIRD YEAR																			
4. FOURTH YEAR & BEYOND																			
B. UNCLASIF. TOTAL																			
1. UNDERGRAD																			
2. POSTGRAD																			
C. 1ST PROF STAGE																			
1. GRAD STUD TOTAL																			
2. FIRST YEAR																			
3. ALL OTHER																			
II. PART-TIME TOTAL																			
A. UNDERGRAD TOTAL																			
1. FIRST YEAR																			
2. SECOND YEAR																			
3. THIRD YEAR																			
4. FOURTH YEAR & BEYOND																			
B. UNCLASIF. TOTAL																			
1. UNDERGRAD																			
2. POSTGRAD																			
C. FIRST PROF																			
1. GRAD STUD TOTAL																			
2. ALL OTHER																			
III. GRAND TOTAL ALL																			

MOVEMENT FROM HARDCOPY REPORTING

In an attempt to make enrollment information more easily accessible to the UCF community and the general public, the IR office began creating MS Excel files for display on the web. Thus, each semester when the reports are printed, IR's webmistress manually creates spreadsheets that could then be accessed from the IR website and downloaded to a local machine. Figure 2 shows the web display of the report shown in Figure 1. A PDF file is also created manually for ease of printing. This process is repeated twice for the fall and spring semesters, as first a preliminary report is created and then a final report, and once for the summer semester. Although the IR office is now able to provide enrollment information to the masses, it is an extremely labor-intensive process to maintain these types of data sets and static web reports. Also, the type of information provided is limited; therefore, not all of the university's data needs are being met. Thus, what we need is a true dynamic web-based reporting environment.

Figure 2. MS Excel Web View of Headcount Report

DYNAMIC WEB-BASED ENVIRONMENT

The decision was made to advance to a dynamic environment that provides information on student headcount enrollments in a myriad of different ways. For example, this application shows enrollments by college, undergraduate/graduate, full-time/part-time status, gender, ethnicity, classification and major in a drill-down fashion. For new students, headcount by student type is available at either the undergraduate or graduate level. This website was designed to replace and enhance numerous hard copy reports and be accessible via the web in a user-friendly, dynamic and interactive environment.

DATA ACQUISITION PROCESS

Using an Open DataBase Connectivity (ODBC) connection and a tool named Shadow Direct to access the student enrollment data on the mainframe, the data is brought into an MS Access

database. Then SAS/ACCESS to PC File Formats is used to import the data into a permanent SAS data set. For the current term this process is repeated daily using "Windows Scheduled Tasks" to run two batch programs. The first program runs an MS Access macro to update the database, whereas, the second program runs a SAS program to import the data for use by the application. Historical data is captured in a similar fashion each time a new file is submitted to the state board of education.

RunAccess.bat

```
"C:\Program Files\Microsoft Office\Office\MSACCESS.EXE"
"C:\Enrollment\Database\Enrollment.mdb /x Student
```

RunSAS.bat

```
"C:\Program Files\SAS Institute\SAS\V8\sas.exe" -sysin
"C:\Enrollment\Database\importstudent.sas
```

import.sas

```
options sasautos='C:\Enrollment';
libname enroll 'C:\Enrollment\Database';
PROC IMPORT OUT= ENROLL.SUMMER03
DATATABLE= "Coded Students"
DBMS=ACCESS2000 REPLACE;
DATABASE="C:\Enrollment\Database\
Enrollment.mdb";
```

RUN;

DEMONSTRATION

The following screen shots are a small representation of what is available on the Enrollment Profile site. Some of the SAS programs that produce these pages are very lengthy and, therefore, will not be reproduced in their entirety. Samples of code will be provided for key items or ideas.

Figure 3 shows the home page of the Enrollment Profile site. Instead of an HTML form, a SAS program is used to calculate the percentage difference and display a horizontal bar chart created using the Java device driver. The macro variables used in the program are passed in the URL as name/value pairs: http://www.irweb2.ucf.edu/scripts/broker.exe?_service=beta&_program=enroll_headcount.sas&_debug=0. A portion of "headcount.sas" shows how the options are set for SAS/GRAPH.


```
/*Set options for graph output device*/
options reset=all
device=java
border
xpixels=525
ypixels=110
cback=white;

/*Set colors for graphs*/
pattern1 c=cx0000A0;
pattern2 c=cx800040;
```

```
/*Set axis options*/
axis1 label=none minor=(n=4);
axis2 label=none;

/*Tell SAS to output chart to the web*/
ods listing close;
ods html body= webout rs=none
archive="/sasweb/graph/graphapp.jar"
attributes=("codebase="/sasweb/graph")
parameters=("userfmt1"="value $totfmt
'total1'=&tm &year 'total2'=&tm &preyr");
```

Figure 3. Home page of Enrollment Profile



Clicking on the “Summer 2003 Detailed Information” button brings you to the page shown in Figure 4. If you want to view information from a previous term and/or year, clicking on “Select Another Semester/Year” provides that option. On this page there are active links to drill-down deeper into the information. You can view trends as in Figure 5 by selecting a link under the “Trends” heading, or clicking a link under the “College” heading will display Figure 6, which breaks down the headcounts by classification, gender, undergraduate/graduate, and full-time/part-time status.

The following code makes the dynamic links by creating a format for the “Trends” and “College” variables that will be used later in the PROC TABULATE statements.

```
data acadfmt;
  length label $200;
  set temp(keep=aca acadgrp college);
  retain fmtname 'ACADFMT' type 'C';
  start=aca;
  label=trim('<A
  HREF="http://www.irweb2.ucf.edu/scripts/broke
  r.exe?'"||
  ' service=beta&_debug=0&_program=enroll.coll_
  trends.sas' ||
  '&new=' || "&new" ||
  '&term=' || "&term" ||
  '&hon=' || "&hon" ||
  '&year=' || trim(left(&year)) ||
  '&college=' || trim(college) || ">" ||
  trim(acadgrp) || '</A>');
run;
```

```
proc format cntlin=ACADFMT;
run;
quit;
data collfmt;
  length label $200;
  set temp(keep=col college college_name);
  retain fmtname 'COLLFMT' type 'C';
  start=col;
  label=trim('<A
  HREF="http://www.irweb2.ucf.edu/scripts/broke
  r.exe?'"||
  ' service=beta&_debug=0&_program=enroll.class
  _lev_time.sas' ||
  '&new=' || "&new" ||
  '&term=' || "&term" ||
  '&hon=' || "&hon" ||
  '&year=' || trim(left(&year)) ||
  '&college=' || trim(college) || ">" ||
  trim(college_name) || '</A>');
run;
```

```
proc format cntlin=COLLFMT;
run;
quit;
```

The following portion of code tells SAS to close the output window and send the procedure output to the web browser.

```
ods listing close;
ods html body=_webout (notop robot)
path=&_tmpcat (url=&_replay) rs=none;
```

The following is a portion of the PROC TABULATE code showing the format statement.

```
proc tabulate data=temp format=comma8.
  . . . other SAS statements
  format col $collfmt. aca $acadfmt.;
  . . . other SAS statements
run;
```

Then we need to stop the output to the browser and re-open the output window.

```
ods html close;
ods listing;
```

Figure 4. Headcount by College





Figure 5 displays enrollment trends for summer semesters over the 5-yr period from 1998 to 2002. The Java device driver was used to create this grouped vertical bar chart.

Figure 5. College 5-yr Headcount Trend



There are two sets of active links in Figure 6. The blue links take you to the output shown in Figure 7 and the black links go to the glossary where an explanation is given of how we define the fields for this site.

Figure 6. Headcount by Classification, Gender and Enrollment Status




Figure 7 has dynamically generated links that will display the most recent 5-yr trend for the chosen major when clicked (Figure 8).

Figure 7. Headcount by Major, Gender, and Enrollment Status






Figure 8. Major 5-yr Headcount Trend



The drop-down menu displayed in Figure 9 is created using JavaScript and gives the user immediate access to a particular view without having to drill-down from page to page. Thus, the functionality is interactive—the users have more control over the


data they specifically want to see. Each page (except the home page) displays this drop-down menu; thus, a separate macro program containing the script was written rather than repeat the same lengthy code in multiple programs.

Figure 9. Interactive Drop-Down Menu



As shown in Figure 10, the same data displayed in Figure 1, which was manually entered into an MS Excel spreadsheet, is now dynamically generated and displayed by the click of a hyperlink.

Figure 10. Dynamically Generated Version of Figure 1



COURSE REGISTRATION PROFILE

Schedulers and course planners at UCF need a way to track daily course registration such as what courses are being offered, what the daily enrollments in courses are, who the instructor is, etc. in order to provide the university with the appropriate offering of classes. This information changes on a daily basis during the registration period but the previous day's data was overwritten. Thus, there was not a way to track registrations over time, which would facilitate a decision to open up a new section if needed.

HARDCOPY REPORTING

Legacy reporting can be challenging to work with. Many of our legacy applications used Customer Information Control System (CICS) screens, as shown in Figure 11, that are difficult to change or redesign the views. A specific example is the field length for FIN ENR is limited to three digits and we now have web-based courses with enrollments over 1000, so the first digit is chopped off in the display. Another difficulty is that there is no search functionality and it can be cumbersome to find information on the screen.

Figure 11. Master Schedule Information

TERM	PRFX	CRS. NBR	T	S	SEC. NBR	REG. KEY	COURSE TITLE	CR. HRS.	CRS. DEPT	INSTR. NAME	CRS. FIN LMT	ENR
00208	ACG	2021			0001	6987	PRIN FIN ACCTNG	003.0	1320	VEIT M R	072	068
00208	ACG	2021			0002	6416	PRIN FIN ACCTNG	003.0	1320	VEIT M R	068	057
00208	ACG	2021			0004	6418	PRIN FIN ACCTNG	003.0	1320	EVANS T G	295	255
00208	ACG	2021			0005	6419	PRIN FIN ACCTNG	003.0	1320	SALTER J H	295	228
00208	ACG	2021			0006	6420	PRIN FIN ACCTNG	003.0	1320	SALTER H P	072	069
00208	ACG	2021			0007	6421	PRIN FIN ACCTNG	003.0	1320	SALTER H P	072	067
00208	ACG	2021			0008	6422	PRIN FIN ACCTNG	003.0	1320	SALTER J H	072	067
00208	ACG	2021			0001	6428	PRIN MGL ACCTNG	003.0	1320	THORNBERG	063	050
00208	ACG	2021			0002	6429	PRIN MGL ACCTNG	003.0	1320	THORNBERG	059	049
00208	ACG	2021			0003	6430	PRIN MGL ACCTNG	003.0	1320	SMITH V S	072	059
00208	ACG	2021			0004	6431	PRIN MGL ACCTNG	003.0	1320	SMITH V S	072	069
00208	ACG	2021			0005	6432	PRIN MGL ACCTNG	003.0	1320	PAVILONIS	055	058
00208	ACG	2021			0006	6433	PRIN MGL ACCTNG	003.0	1320	SALTER H P	072	043
00208	ACG	2021			0007	6434	PRIN MGL ACCTNG	003.0	1320	SALTER H P	072	069
00208	ACG	2021			0008	6435	PRIN MGL ACCTNG	003.0	1320	THORNTON T	058	058
00208	ACG	2021			0010	6437	PRIN MGL ACCTNG	003.0	1320	PAVILONIS	072	085

MOVEMENT FROM HARDCOPY REPORTING

After the university went live with a PeopleSoft® Student Administration system, course planners and other people in the colleges and departments were having an extremely difficult time getting any reports containing course schedule information. A web-based class schedule search application was created (Figure 12), but it didn't provide all of the information that was available on the old CICS screens. At the same time, Dr. Taylor Ellis, Associate Dean of Undergraduate Programs in the College of Business, was developing a static HTML application using SAS for his own use to track course registrations. The programs were run manually and the pages stored on a local machine. Each semester, course information was manually entered to create the hyperlinks, which was labor-intensive for the programmer. What was needed was a tool that would allow anyone access to key registration data, give users control over what type and level of information they wanted to see, and not require extensive maintenance time.

Figure 12. Class Schedule Search Application

Key Code	Subject	Catlg No.	Title	Section	Comp	Status
52624	ACG	2021	PRIN FIN ACCTNG	A001	LEC	Closed
52683	ACG	2021	PRIN FIN ACCTNG	B001	LEC	Closed
52494	ACG	2021	PRIN FIN ACCTNG	A001	LEC	Open
52495	ACG	2021	PRIN FIN ACCTNG	A002	LEC	Open
52496	ACG	2021	PRIN MGL ACCTNG	A001	LEC	Closed
52497	ACG	2021	PRIN MGL ACCTNG	A002	LEC	Closed
52498	ACG	2021	PRIN MGL ACCTNG	A003	LEC	Open

DYNAMIC WEB-BASED ENVIRONMENT

The Course Registration Profile is a web-based application that provides information on course registration based on criteria selected by the user. Graphical and tabular reports show how quickly classes are filling up by campus, college, course modality, undergraduate/graduate, or specific courses(s). Users can drill-down from course prefix to course number to course section. At the course and section level, the user can access a plot that shows course enrollment over time. As time progresses, we would be able to create historical data files for future trend comparisons. Much of the same technology used in the Enrollment Profile, specifically SAS/IntrNet, is used to dynamically generate the pages and hyperlinks.

DATA ACQUISITION PROCESS

This application needs to display more up-to-date information so it was decided to access the PeopleSoft data directly, which is

stored in Oracle tables, using SAS/ACCESS to Oracle technology. During peak registration times the data would be refreshed 4 times daily so a "Windows Scheduled Task" was created with multiple running times to run a SAS program. Throughout the day the data are replaced and the data from the last run are saved and stored.

RunSASPS.bat

```
"C:\Program Files\SAS Institute\SAS\V8\sas.exe" -sysin
C:\CourseReg\PSView\Database\getdata.sas
```

getdata.sas

```
%macro getdata(daily,all,tmid);
... other SAS statements

/*SAS/ACCESS to ORACLE*/
libname ps oracle user=userid
password=password path='data source name';
proc sql;
create table psreg.&daily as
select * from ps.PS_CF_CLASS_IR_VW
where STRM=&tmid and CLASS_NBR ne . and
CLASS_STAT in ('A','T','S') and
COMPONENT in
('DIS','LAB','LEC','PER','SEM') and
substring(CATALOG_NBR from 2 for 3)
not in ('903','904','905','906',
'907','908','909','912','917','918',
'919','940','941','944','946','949',
'957','958','970','971','973','980');
quit;
libname ps clear;
... other SAS statements
```

DEMONSTRATION

Since it was not required to display graphs or provide calculations on the home page of the Course Registration Profile a standard HTML form is used to collect the name/value pairs needed to begin the application. Figure 13 shows the home page where you can select filters for the information desired. If information for a specific course is desired, ENC1101 for example, a text box is present to type in your request.

Figure 13. Home Page of Course Registration Profile

The screenshot shows a web browser window with the URL 'http://www.uvwi2.ucf.edu/reg/PSView/CourseReg1.html'. The page title is 'Course Registration Profile' and it features a search form with the following fields:

- Campus:** A dropdown menu with 'All Campus Locations' selected.
- College:** A dropdown menu with 'All Colleges' selected.
- Instructional Mode:** A dropdown menu with 'All Instructional Modes' selected.
- Course Level:** A dropdown menu with 'All Levels' selected.
- Specific Course(s):** A text input field with a placeholder 'Example: ENC1101 or ENC1101H'.

 There are also several images of the University of Central Florida campus and a link to 'Open Registration Calendar in a new window'.

After clicking the submit button on the home page, the next page (Figure 14) has options to filter the data for a specific department based on the college selected, term and year selections, an option to exclude the display of Honors courses, and the ability to see information for courses where enrollment is at a certain percentage.

Figure 14. Department and Term Selection




Figure 15 shows the course registration information by course prefix. SAS/GRAPH generates a pie chart using the GIF driver and the REPORT procedure generates the tabular display. Dynamic hyperlinks to drill-down to the next level are generated as detailed above in the Enrollment Profile. The ANNOTATE facility is used to generate custom displays for the pie chart. The ODS MARKUP destination (experimental in Version 8.2) is used to provide ALT information for the graph to comply with accessibility standards. There is a hyperlink available to display all the sections in a list format and another to go to the class schedule search shown in Figure 12.

```
proc template;
  define tagset tagsets.test;
    parent=tagsets.html;
    define event image;
      put "<img";
      put " border=2 alt=""SAS
          Pie Chart""";
      putq " src=" URL ">" NL;
    end;
  end;
run;

ods listing close;
ods markup tagset=test body=webout (notop
notob) path=&_tmpcat (url=&_replay)
style=styles.ucf rs=none;

/*Set Graphics Options*/
goptions reset=all
  device=gif260
  gunit=CELLS
  cback=white
  border
  ftext=swissxb
  ftrack=loose
  htext=1.5;

pattern1 value=psolid color=vlib;
pattern2 value=psolid color=CXFBEAB5;


data annotest;
  length function color style $ 8 text $ 25;
  retain function 'label' color 'black' when
    'a' style 'swissxb' xsys ysys '2'
    position '5' hsys '3' size 5.5;
  set temp3;
  if status='sumenr';
  midpoint=status;
  x=50;
```

```
y=50;
text=left(put(ratio,percent8.));
output;
run;

data _null_;
  set temp3;
  if status='sumenr' and ratio>1 then
    call symput('grp','annotate=annotest');
  else
    call symput('grp','percent=outside');
run;

proc gchart data=temp3;
  where number>0;
  format status $statfmt. number comma8.0;
  pie3d status /sumvar=NUMBER
    &grp
    value=outside
    slice=outside
    angle=60
    midpoints='Enrolled'
    'Available'
    nogroupheading
    noheading
    nolegend;
run;
quit;
ods markup close;
ods listing;
```

Figure 15. Course Registration by Course Prefix



The listing of all sections (Figure 16) is convenient for department heads who want to see all of a department's offered courses at once. However, printing an HTML table can sometimes be difficult. Sometimes, key information is chopped off at the bottom or top of the page. A hyperlink is available to dynamically generate a PDF file using the ODS PDF output destination available in SAS Version 8.2 (Figure 17). Another hyperlink is available to serve the temporary data set to the browser, first exported to an MS Excel file that can be saved on the user's local machine for data manipulation (Figure 18). A hyperlink is also available to display a data dictionary for some of the data fields in the table.

```
options orientation=landscape
papersize=letter pageno=1
  leftmargin=.5 rightmargin=.5 topmargin=.5
  bottommargin=.5;
ods listing close;
ods pdf file=webout style=styles2.ucf
startpage=never;
```

```
ods proclabel "Expanded Course Listing";
```

```
%let rc=%sysfunc(appsrv_session(create));
PROC EXPORT DATA=WORK.TEMP
OUTFILE="C:\CourseReg\PSView\
temp&_sessionid..xls"
DBMS=EXCEL2000 REPLACE;
RUN;
put 'a
href="http://www.irweb2.ucf.edu/scripts/broke
r.exe? service=beta& program=sashelp.webprog.
filesrv.scl& filetype=e& debug=0& file=c:\cour
sereg\psview\temp&_sessionid..xls"
target="blank">';
```

Figure 16. Expanded Listing of All Sections




Figure 17. Dynamically Generated Acrobat PDF






Figure 18. Serve Excel File to Web



Clicking on one of the prefix hyperlinks in Figure 15 will display registration information by course number for each of the courses beginning with the prefix selected (Figure 19).

Figure 19. Course Registration by Course Number



The TEMPLATE procedure is used to modify the style for the tabular display of the REPORT procedure. The following macro is called prior to the code that produces the table and the style declaration is added to the ODS HTML statement.

```
%macro style1;
/*Create style sheet for browser output*/
ods path work.templat(update)
sashelp.templmst(read);
proc template;
define style Styles.UCF;
parent = styles.default;
replace fonts
"Fonts used in the default style" /
'FooterFont' = ("Verdana, Arial,
Helvetica, Helv",2,Bold)
'TitleFont2' = ("Verdana, Arial,
Helvetica, Helv",2,Bold Italic)
'TitleFont' = ("Verdana, Arial,
Helvetica, Helv",3,Bold Italic)
'StrongFont' = ("Verdana, Arial,
Helvetica, Helv",2,Bold)
'EmphasisFont' = ("Verdana, Arial,
Helvetica, Helv",1,Italic)
'FixedEmphasisFont' =
("Courier",1,Italic)
'FixedStrongFont' =
("Courier",1,Bold)
```

```

'FixedHeadingFont' = ("Courier",1)
'BatchFixedFont' = ("SAS Monospace,
  Courier",1)
'FixedFont' = ("Courier",1)
'headingEmphasisFont' = ("Verdana,
  Arial, Helvetica, Helv",2,Bold
  Italic)
'headingFont' = ("Tahoma, Arial,
  Helvetica, Helv",2,Bold)
'docFont' = ("Tahoma, Arial,
  Helvetica, Helv",2);
replace color_list
"Colors used in the default style" /
'fgB2' = cx0000ff
'fgB1' = cx800040
'fgA4' = cx000000
'bgA4' = cxffffff
'bgA3' = cxffffff
'fgA2' = cx000000
'bgA2' = cxffffff
'fgA1' = cx000000
'bgA1' = cxffffff
'fgA' = cx000000
'bgA' = cxffffff;
replace colors
"Abstract colors used in the default
  style" /
'headerfgemph' = color_list('fgA2')
'headerbgemph' = color_list('bgA4')
'footerfgstrong' =
  color_list('fgA2')
'footerbgstrong' =
  color_list('bgA4')
'headerfgstrong' =
  color_list('fgA2')
'headerbgstrong' =
  color_list('bgA4')
'headerfg' = color_list('fgA2')
'headerbg' = color_list('bgA2')
'datafgemph' = color_list('fgA1')
'databgemph' = color_list('bgA3')
'datafgstrong' = color_list('fgA1')
'databgstrong' = color_list('bgA3')
'datafg' = color_list('fgA1')
'databg' = color_list('bgA3')
'batchfg' = color_list('fgA1')
'batchbg' = color_list('bgA3')
'tableborder' = color_list('fgA1')
'tablebg' = color_list('bgA1')
'notefg' = color_list('fgA1')
'notebg' = color_list('bgA')
'bylinefg' = color_list('fgA2')
'bylinebg' = color_list('bgA')
'captionfg' = color_list('fgA1')
'captionbg' = color_list('bgA')
'proctitlefg' = color_list('fgA')
'proctitlebg' = color_list('bgA')
'titlefg' = color_list('fgA')
'titlebg' = color_list('bgA')
'systitlefg' = color_list('fgA')
'systitlebg' = color_list('bgA')
'Conentryfg' = color_list('fgA')
'Confolderfg' = color_list('fgA')
'Contitlefg' = color_list('fgA')
'link2' = color_list('fgB2')
'link1' = color_list('fgB1')
'contentfg' = color_list('fgA2')
'contentbg' = color_list('bgA2')
'docfg' = color_list('fgA')
'docbg' = color_list('bgA');
style Body from Body /

```

```


pagebreakhtml =
  %nrstr("<p style=""page-break-after:
    always;"">&#160</p><HR size=2>");
style SystemFooter from SystemFooter /
  foreground = black
  font = fonts('footerFont');
style FooterStrong from Footer /
  background =
    colors('footerbgstrong');
replace Output from Container
  "Abstract. Controls basic output
  forms." /
  background = colors('tablebg')
  rules = GROUPS
  frame = BOX
  cellpadding = 7
  cellspacing = 1
  bordercolor = colors('tableborder')
  borderwidth = 1;
replace RowHeader from Header
  "Controls row headers." /
  foreground = color_list('bgA1');
end;
run;
%mend style1;

ods html body=_webout (notop robot)
  style=styles.ucf
  rs=none;

```

Selecting a course number hyperlink displays each section offered as shown in Figure 20. Clicking the "Course:" hyperlink will produce Figure 21.

Figure 20. Course Registration by Course Section



To change the background color of a cell in the REPORT procedure if certain conditions are met we define a compute variable named "FYI" and then provide the if-then logic.

```

compute FYI /character length=40;
if (_C6_ < _C7_) and (_C7_ < _C8_) then
do;
FYI = "RmCap < ClsLim < RegLim";
call define(_COL_, "STYLE",
  "style(CALLDEF)=
  {background=CXD8D8D8
  foreground=black}");
call define('_C6_', "STYLE",
  "style(CALLDEF)=
  {background=CXD8D8D8
  foreground=black}");
call define('_C7_', "STYLE",

```



```

"style(CALLDEF)=
{background=CXD8D8D8
foreground=black}");
call define('_C8_', "STYLE",
"style(CALLDEF)=
{background=CXD8D8D8
foreground=black}");
end;
else if _C6_ < _C7_ then
do;
similar statements as above
end;
else if _C8_ < _C9_ then
do;
similar statements as above
end;
else
FYI="";
endcomp;

```

PROC Gplot is used to produce the plot in Figure 21 and the ANNOTATE facility is used to put the date of registration below the horizontal axis and to display the numbers above each point.

```

data annotest;
length function color style $8 text $10
position $1;
retain function 'label' when 'a' xsys '2'
hsys '3' position '6';
set temp;
if number>0 &shw;

/*Put dates on axis*/
style="Arial"; color='black'; x=day;
y=4; text=put(prev,mmddyy8.);
ysys='3'; angle=90; size=&sz; output;

/*Label data points with numbers*/
style='swissb'; color='blue'; x=day;
y=number; text=put(number,5.);
ysys='2'; angle=90; size=&dt; output;

run;

```

A hyperlink is provided to compare the course registration with the previous year (Figure 22). Two plots, one for the current year and one for the previous year, are shown using the OVERLAY option of the PLOT statement. Clicking on "Registration Calendar" will display Figure 23 and clicking on a course section hyperlink shown in Figure 20 will produce a plot of course registration over time for that section (not shown).

Figure 21. Course Registration Over Time for a Specific Section




Figure 22. Trend Comparison with Previous Year




Figure 23. Registration Calendar

Appointment Times for Summer 2003			
Special Handling (RSPC)	Undergraduates	24-Mar	9:00am-7:00pm
	Graduates	24-Mar	8-8:30am
Graduates	Doctoral (RAL1)	25-Mar	7:00am-8:00am
	Specialist (RAL2)	25-Mar	8:30am
	Masters (RAL3)	25-Mar	9:00am-5:00pm
Seniors (RAL6)		25-Mar	6:00pm-10:00pm
		26-Mar	6:00am-10:00pm
Second Degree (RAL5)		27-Mar	2:00pm-3:00pm
		27-Mar	4:00pm-10:00pm
Juniors (RAL7)		28-Mar	6:00am-7:00pm
		28-Mar	2:00pm
Transfer Orientation			
Seniors/Juniors w/o CLAST (RCNM)		31-Mar	10:00am-1:00pm
New Degree Seeking Grads		31-Mar	9:00am
Sophomores (RAL8)		31-Mar	2:00pm-8:00pm
		1-Apr	9:00am-5:00pm
Freshman (RAL9)		2-Apr	9:00am-8:00pm
Post-Bacs (RAL4)		3-Apr	9:00am-12:00pm

CONCLUSION

Constituents at UCF have a strong desire for dynamically accessible data. New applications were developed to meet user needs, prototyped to the University of Central Florida community, and extremely well received. SAS technology, specifically SAS/IntrNet, has provided the IR office with the tools needed to deliver timely and accurate data in a user-friendly dynamic web-based reporting environment. Once the initial programs are built, there is much less programmer time involved to maintain these types of applications. The IR office intends to continue the movement using SAS to develop administrative applications and serve as a leader in this area of database design and application.

REFERENCES

The following SAS publications were invaluable tools used extensively during the creation of these two applications.

SAS Institute Inc. (1998), *SAS[®] Macro Language, Course Notes*, Cary, NC: SAS Institute Inc.

SAS Institute Inc. (2000), *SAS[®] Web Tools: Advanced Dynamic Solutions Using SAS/IntrNet Software, Course Notes*, Cary, NC: SAS Institute Inc.

SAS Institute Inc. (2000), *SAS[®] Web Tools: Static and Dynamic Solutions Using SAS/IntrNet Software, Course Notes*, Cary, NC: SAS Institute Inc.

ACKNOWLEDGMENTS

We would like to thank the technical support team at SAS for the assistance provided, especially when things were not working as expected.

TRADEMARK CITATION

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

CONTACT INFORMATION

Your comments and questions are valued and encouraged.
Contact the authors at:

Sabrina Andrews
Office of Institutional Research
University of Central Florida
P.O. Box 160021
Orlando, FL 32816-0021
Work Phone: (407) 823-5061
Fax: (407) 823-4769
Email: slandrew@mail.ucf.edu
Web: <http://www.iroffice.ucf.edu>

Evangeline Collado
Office of Institutional Research
University of Central Florida
P.O. Box 160021
Orlando, FL 32816-0021
Work Phone: (407) 823-5061
Fax: (407) 823-4769
Email: ecollado@mail.ucf.edu
Web: <http://www.iroffice.ucf.edu>

Patricia Ramsey
Office of Institutional Research
University of Central Florida
P.O. Box 160021
Orlando, FL 32816-0021
Work Phone: (407) 823-5061
Fax: (407) 823-4769
Email: ramsey@mail.ucf.edu
Web: <http://www.iroffice.ucf.edu>